

# DYNAMIC MODELLING WITH ECOSIMPRO OF LIQUOR FILTERS IN THE SUGAR INDUSTRY

Susana Pelayo Díaz

Centro de Tecnología Azucarera, Universidad de Valladolid  
C/Real de Burgos. Edificio Alfonso VIII. Planta baja s/n 47011 Valladolid - España  
Tfno: 983 42 35 63. FAX: 983 42 36 16. Email: susana@cta.uva.es

## Summary

*The carbonated liquor filters in a sugar factory are complex systems that operate discontinuously. Ecosimpro has been used to model each element of the filter separately. These elements were then assembled to simulate the complete filter, including the parameters and variables that allow the simulation of the batch behaviour.*

**Key Words:** Simulation, EcosimPro, object-oriented modelling, liquor filters, batch processes.

## 1 INTRODUCTION

In a sugar mill, cleaning is carried out in order to eliminate the largest possible amount of extracted impurities (non-sugar) during the diffusion process. The different types of non-sugars are eliminated by successive precipitations at various pH and temperature levels. Improper cleaning intensifies the colour of the sugar obtained and reduces its final quality.

The liquor from the diffusion undergoes several stages where milk of lime is added to obtain certain alkalinity levels and thus precipitate the substances that need to be eliminated. CO<sub>2</sub> is then added so that it reacts with the lime in the liquor and precipitates as calcium carbonate (carbonation), which carries the non-sugar substances precipitated in previous stages. Filters separate these lumps from the liquor. There are two successive carbonation-filtering stages in a typical sugar process.

Filtering is a separation process whereby a liquid flows through a porous membrane that retains the solid elements in the liquid that need to be separated. Sugar factories use the 'filter cake formation' process: the filter establishes the initial filtering characteristics, but as the solids accumulate on its surface, the cake that is formed has an increasing influence on the filtering process.

The duration of a filtering cycle depends on how quickly the pressure on the mesh is increased and, consequently, on how quickly the inlet flow is reduced.

The main variables that influence the filtering process are viscosity, thickness and porosity of the cake, differential pressure and filter surface.

There are also other variables to be considered, although they have less influence: density of the liquor, non-sugar concentration, temperature, duration of the filtering cycle, size and compressibility of the solid particles, addition of filter aids and state of the filter mesh.

The filters of a sugar factory are batch processes and therefore operate discontinuously. A batch process can be defined as a process that leads to the production of finite quantities of material by subjecting quantities of input materials to an ordered set of processing activities over a finite period of time using one or more pieces of equipment (definition of the ISA-S88 standard, 1995).

## 2 PURPOSE

The purpose of this paper can be summarised in the following points:

- study the process unit and the elements that make it up
- study the discontinuous operation of the filter and its modelling
- develop a generic dynamic mathematical model and its simulation with Ecosimpro
- estimate parameters and validate the model by comparing it with real operation data

This work forms part of a project aimed at the development of a complete sugar mill simulator for personnel training.

## 3 DESCRIPTION OF A FILTER

The diagram of the liquor cleaning filter chosen for the model can be seen in figure 1.

This kind of filter has several elements:

- *Filter body:* where the filtering meshes are located
- *Cone-shaped lower base:* where the sludge is stored until it is discharged

- *Blowing storage*: where the first filtered liquor is stored for its later use for mesh cleaning
- *Valves*: allow the inlet and outlet of the various currents and ease cyclic operation of the filter




Figure 1. Liquor filter diagram




Figure 2. Filtering cycle stages

This filtering process has the following generic stages (figure 2):

- 1) *Initial filtering*: as the liquor flows through the mesh, the cakes start forming. This first liquor, which may carry some solid elements if the cake is not completely formed, is kept in the blowing storage for use at a later stage.
- 2) *Filtering*: filtering continues, but the liquor is sent to the filtered liquor storage.
- 3) *End of the filtering*: the filter is decompressed once no cloudy liquor enters it. The liquor inside the blowing storage reenters the filter, but this time in the reverse direction to facilitate the breaking off of the cakes.

The removal of the sludge from the previous cycle is performed.

- 4) *End of the removal*: the cakes that break off accumulate at the bottom of the filter until they are removed during the following cycle.

The stages of each filtering cycle are determined by the timing of the filter (which controls the opening and closing of the existing valves). The programmed timing in this case can be seen in table 1, and its graphic representation is shown in figure 4:

<i>Time</i>	<i>Description</i>
TAUF	Complete cycle time: from the closing of the VA supply valve until the closing of the VA supply valve.
TI	Recycling time: from the closing of the VD pressure relief valve until the closing of the VR recycling valve.
T2	time between the closing of the VA supply valve and the opening of the VD pressure relief valve
T3	Pressure relief time: from the opening of the VD pressure relief valve until the closing of said VD valve
T4	Maximum extraction time: from the opening of the VE extraction valve until the closing of said VE valve
T5	Pre-levelling time: from the opening of the VD pressure relief valve until the opening of the VN levelling valve

Table 1. Filter timing




Figure 3: Visualisation of the filtering cycle stages.

The cloudy liquor entering the filters contains a large number of substances. The following chemical substances have been considered in the model developed: water, sugar, dissolved impurities, precipitated impurities, dissolved calcium, calcium oxide, calcium carbonate, ion carbonate, protons and hydroxyls. Furthermore, dissolved impurities can be considered as a single component or can be divided into proteins, amides, invert sugar and other substances. The model generated allows the use of the value of a parameter so as to treat the impurities as a whole or specifically.

In these filters, the sludge generated in the previous filtering cycle is accumulated at the bottom of the filter in the current cycle. During this time, the heaviest substances are sedimented, so the density of the sludge before the discharge is much greater at the bottom of the filter than at the top.

In view of this fact, each filter has two options to extract the sludge: extraction by 'deltaP' or by 't4 time'. For the extraction by density, the closing mechanism closes valve VE automatically when the set value is reached (desired sludge density). The model only takes into consideration the discharge of the sludge by time.

#### 4 PHYSICAL-MATHEMATICAL MODEL

Conceptually, the filter has been divided into a series of simple elements in order to facilitate modelling (see figure 3):

- blowing tank
- filter mesh

- cone 1: location where the sludge is initially collected after the cakes have broken off
- cone 2: location for final storage of the sludge until it is discharged in the following cycle
- valves




Figure 4: Diagram of components used for the filter model.

The models of each element are based on the conservation laws for matter, energy and amount of movement. A compromise is always sought between the faithful representation of the real process and the mathematical complexity deriving from it, and this compromise implies the consideration of certain hypotheses and the calculation of certain parameters.

The inlet and outlet flows for each element depend on the operating conditions and the state of the filter, which is determined by the moment of filtering and the timing defined. This timing has been included in the model by means of a series of parameters that take on the values of 1 or 0 depending on whether they are activated or not. The activation of these parameters means the automatic valves of the filters are open or closed.

The most relevant aspects of the models for each equipment item are included below:

#### 4.1 FILTER MESH

The solid elements precipitated in the previous stages of the cleaning process accumulate in the filter mesh. Ideal filtering would mean that all solid elements entering the filter would be retained, but in reality a part of the solid elements may flow through the filtering current mainly because of the faulty formation of the cake or breakage of the mesh. A filtering efficiency factor has been included in the model to simulate these effects.

The retention of solid elements causes a gradual increase in the mass of the cake as cloudy liquor flows in (stages 1 and 2 of the mesh), while there is a reduction during the discharge stage (stage 3). In order to ease modelling, it has been considered that only solid elements are retained (precipitated impurities, lime oxide and lime carbonate), and internal variables, known as 'W[j]<sub>RET</sub>', have been defined. These variables refer to the flow of each solid element incorporated to or removed from the cake. During the cake break-up stage, these internal variables have been defined proportionally to the amount of liquor entering from the blowing storage and to the existing mass of each component at each moment.

$$\frac{d\text{mass}_{RET}[j]}{dt} = (k[1] + k[2] - k[3]) \cdot Wi[j]_{RET} \quad (1)$$

$$Wi[j]_{RET} = (k[1] + k[2]) \cdot Wi[j]_{cloudyliquorinlet} \cdot ff + k[3] \cdot \frac{\max(0, \text{mass}_{RET}[j]) \cdot k_{desc} \cdot W_{in. liq. from blow. tank}}{\text{mass}_{RET}[j] + 10^{-8}} \quad (2)$$

where: ff (filtering efficiency), k (parameter that sets the changes of stage),  $k_{desc}$  (proportionality coefficient),  $\text{mass}_{RET}$  (mass of each cake component),

W (total mass flow),  $Wi_{RET}$  (mass flow of each component entering/leaving the cake).

#### 4.4.1 Inlet and outlet flows

The inlet flow to the mesh depends on two factors:

- *Thickness of the cake that develops*: the thicker the cake, the greater the resistance to flow, so the lower the inlet flow. This dependency has been modelled as the result of a maximum rated flow entering a completely clean filter and an exponential correction flow with a value which is reduced as the thickness of the cake increases.
- *Control signal from the level controller of the cloudy liquor storage tank*. Operation of the level controller would affect the common pump that supplies all the filters, and would therefore also affect the pressure difference resulting from the filter inlet flow. In the model, the effect of the outlet of the controller has been applied directly on the inlet flow for each filter. This simplification prevents the appearance of complex algebraic loops that significantly slow down the simulation of all the cleaning section.

$$W_{cloudyliquorinlet} = s_{in} \cdot [(k[1] + k[2]) \cdot \exp(-\sqrt{esp} \cdot f_{esp}) \cdot W_{max}] \quad (3)$$

At the start of the filtering cycle, the liquor obtained as filtered is sent to the blowing storage tank, where it is kept until the cakes are later discharged. After a certain filtering period, once the thickness of the cake is enough to ensure proper retention of solid elements, the liquor obtained as filtered flows to the filtered liquor storage tank and from there to the following process stages. The total liquor flow in both stages is the inlet flow plus the flow retained in the mesh.

$$W_{outletliqortoblowingtank} = k[1] \cdot (W_{inletcloudliquor} - W_{RET}) \quad (4)$$

$$W_{filteredliquor} = k[2] \cdot (W_{inletcloudliquor} - W_{RET}) \quad (5)$$

The composition of these liquors must be recalculated on the basis of the substances retained at the cake.

During the cake discharge stage, the liquor accumulated in the blowing storage tank enters as reverse flow, thus facilitating the evacuation of the cake. The mixture of this liquor and the cake (known as sludge because of its large solid content) is sent to cone 1. The total sludge flow obtained is defined by the following equation:

$$W_{outletsludge} = k[3] \cdot (W_{inletslugblowingtak} + W_{RET}) \quad (6)$$

Where: esp (thickness of the cake),  $f_{sp}$  (correction factor), k (parameter that establishes the change of stage),  $s_{in}$  (control signal), W (total mass flow of each flow),  $W_{max}$  (maximum mass flow entering the filter),  $W_{RET}$  (total mass flow entering/exiting the cake).

#### 4.4.2 Temperature and Pressure in the Mesh

Filtering is a separation process with no associated heat flows (with the exception of the small losses towards the exterior). These losses have been disregarded in the model and the process has been considered to be isothermal.

The pressure the liquor has to exert to flow through the mesh increases as the cake becomes thicker. This dependency has been modelled with an exponential correction factor. During the cake decompression and discharge stage, the pressure at the mesh assumes the value defined for this process.

$$P_{mesh} = (k[1] + k[2]) \cdot [P_O + (1 - \exp(-esp \cdot f_{pressure}))] + k[3] \cdot P_F \quad (7)$$

$$P_{outlet1conel} = P + \rho \cdot g \cdot 10^{-5} \cdot (h - (z_1 - z_{bottom})) \quad (10)$$

where: esp (thickness of the cake),  $f_{pressure}$  (correction factor), k (parameter that establishes the change of stage),  $P_F$  (pressure during the decompression stage),  $P_O$  (initial pressure),  $P_{MESH}$  (pressure at the mesh).

#### 4.2 BLOWING TANK

The blowing tank stores liquor that enters during the first stage of the cycle and later returns to the filter to break the cake up. If the amount of liquor entering the tank is greater than its capacity, it spills over and is recirculated to the cloudy liquor tank that supplies the filters.

The model for this equipment has been developed on the basis of the conservation equations, taking into consideration that it is a tank with an inlet and an outlet located at its base and with a overflow line.

#### 4.3 CONE 1

Cone 1 is an intermediate element where the filtering cakes are kept until they can be transferred to cone 2 after discharge of the sludge from the previous cycle. The model for this equipment has been performed on the basis of the conservation equations, considering that it behaves like a stirred tank with an inlet through the top, an outlet through the bottom and a spillover

line at a certain height from the base. In order to simplify the model, the geometry of this element has been likened to a cylinder.

The spillover flow occurs when the height of the sludge accumulated in the cone is greater than the elevation of the spillover line. The timing of the filter (opening and closing of the levelling valve located in that outlet) must also be considered.

$$W_{outlet2conel} = vN\_ON \cdot 1000 \cdot \max(0., h - z_2)^{1.5} \quad (8)$$

The sludge outlet flow through the bottom of the cone has been calculated on the basis of the mass that remains in the cone:

$$W_{outlet1conel} = \max(0., f_{discharge} \cdot mass / 1.) \quad (9)$$

The pressure at the base of the cone is the result of the surface plus the pressure load of the liquid column:

where:  $\rho$  (density of the sludge),  $f_{discharge}$  (correction factor), g (gravity), h (height of the sludge in cone 1), mass (accumulated mass in cone 1), P (pressure), vN\_ON (levelling valve opening/closing parameter), W (total mass flow of each current),  $z_1$ ,  $z_2$ ,  $z_{bottom}$  (height of the outlets and the bottom of the cone).

#### 4.4 CONE 2

The sludge from cone 1 enters cone 2 and is stored there until it is discharged during the following filtering cycle. The model for this equipment item has been developed on the basis of the conservation equations. The geometry has been likened to a cylinder to simplify the model, with an inlet through the top and an outlet through the bottom.

The settling of solid components (precipitated impurities, impz\_p; calcium oxide, CaO; calcium carbonate, CaCO<sub>3</sub>) has to be taken into consideration to calculate the concentration at the moment of the sludge discharge. The modelling of this phenomenon (sludge entering the cone) is performed by assigning the inlet concentration to certain internal variables 'C\_int[j]' and the volume of sludge that entered to the capacity of cone 'V2'.

```
WHEN (fill==1.) THEN
  FOR (j IN ld_mix)
    C_int[j] = f_in.C[j]
  END FOR
END WHEN
```

```
WHEN (End_cycle==1.) THEN
  V2 = V
  a=1.
END WHEN
```

When the discharge occurs, the outlet concentration of these three components must be at its maximum level and be progressively reduced as the sludge flows from the tank. The density of the sludge must therefore be very similar to that of the liquor during the final moments of the discharge.




Figure 5. Distribution of the concentration in cone 2 at the moment of the sludge discharge

A significant simplification of this phenomenon has been made to perform the simulation. It has been considered that there is a linear distribution of the concentration with respect to the height and that the concentration at mid-height (or volume) corresponds to the concentration at the sludge inlet from cone 1 (figure 5).

The concentration at the moment of the discharge is therefore determined by:

- For solids

$$C[j]_{outlet} = (C_{int}[j] \cdot 2 / V2) \cdot V \quad (11)$$

- For the remaining liquid:

$$C[j]_{outlet} = (C_{int}[j] \cdot 2) - (C_{int}[j] \cdot 2 / V2) \cdot V \quad (12)$$

where: C (concentration),  $C_{int}$  (concentration at inlet to cone 2), V (occupied volume in cone 2), V2 (volume initially occupied in cone 2).

#### 4.5 VALVES

The valves in the filter allow the timing of said filter so as to carry out the filtering cycle.

The flow through a valve is proportional to the square root of the pressure difference at the ends of said valve (Bernoulli).

$$W = v\_ON \cdot k \cdot \sqrt{\max(10^{-10}, P_{inlet} - P_{outlet})} \quad (13)$$

where: k (proportionality constant), P (pressure), v\_ON (valve opening/closing parameter), W (mass flow through the valve).

#### 5 DEVELOPMENT OF THE MODEL IN ECOSIM

The mathematical model has to be transferred to Ecosim once it is available. For this purpose, the above-mentioned simple components are created and then joined in a global 'filter' component.

A set of general libraries have had to be developed, apart from the components representing the filter elements. These libraries include.

- chemical components and their physical and chemical properties
- ports
- flow elements: pipes, pumps, valves....
- control elements: regulators, meters, ...
- basic process elements: heat exchangers and tanks.

The code below (corresponding to the liquor port generated for this cleaning section) is presented as an example of these libraries:

```
-----
--- Nombre del Puerto: jugo_depurado.
--- Descripción: Puerto para JUGO o LODO en sección de
DEPURACIÓN.
-----
-- simple=TRUE --> no se especifican impurezas
-- simple=FALSE --> se especifica impurezas
-- ld=TRUE --> el flujo es de lodo
-- ld=FALSE --> el flujo es de jugo
-----
PORT jugo_depurado
  (SET_OF(Chemical)Mix,
  BOOLEAN simple,BOOLEAN ld)

SUM REAL W RANGE 0.,Inf "Flujo másico (Kg/s)"
SUM IN REAL Wi[Mix] RANGE 0,Inf "Flujo
másico de cada componente (Kg/s)"
EQUAL OUT REAL C[Mix] RANGE 0.,1.
```

```

"Concentraciones (% 1 en peso)"
EQUAL REAL P RANGE 0.,Inf "Presión (bar)"
EQUAL OUT REAL T RANGE 0.,Inf
"Temperatura (°C)"
REAL H "Entalpía específica KJ/Kg"
SUM IN REAL f_energ "Flujo de entalpía (KJ/s)"
REAL Pol RANGE 0.,1. "Polaridad en % 1 en peso"
REAL Pureza RANGE 0.,1. "Pureza en % 1 en peso"
REAL Brix RANGE 0.,100. "Grados Brix en %100
en peso"
REAL Rho RANGE 0.,Inf "Densidad (Kg/m^3)"
REAL F
REAL impz_dtas RANGE 0.,1. "impurezas disueltas
totales(% 1 en peso)"
REAL solidos RANGE 0.,1. "fracción de sólidos
totales(% 1 en peso)"

```

#### CONTINUOUS

```

I = SUM (j IN Mix; C[j])
EXPAND(j IN Mix EXCEPT setofElem(Mix,1))
  Wi[j] = C[j] * W
W = SUM(j IN Mix; Wi[j])
Pureza = zona(Pol,max((Brix/100.),0.01),0.)
Pol = C[azucar]
f_energ = W * H
W = F * Rho
EXPAND(simple==TRUE) impz_dtas = C[impz]

EXPAND(simple==FALSE)
  impz_dtas = C[proteinas] + C[amidas] \
    + C[az_inv] + C[otros]

```

```

solidos = solidos_jugoDepurado(Mix,C,ld)
Brix = Brix_jugoDepurado(Mix,C,impz_dtas,ld)
H = entalp_jugoDepurado(Mix,T,C,Brix,Pureza,ld)
Rho = den_jugo(T,Brix,Pureza)

```

ENDPORT

## 6 FILTER COMPONENT

The filter is a component formed by all the above simple components which also has the equations corresponding to the timing of the filtering cycle added to it.

### 6.1. COMPONENT DEFINITION

When calling up the components that make up the filter, the necessary parameters and data to make it specific to the case under study need to be determined.

Some of the parameters receive a value when each component is called up. Others, however, are assigned the corresponding parameter defined in the joint component, so that they acquire a value automatically when the filter parameters are assigned a value.

As regards the data, some of them have been assigned the value of variables or data defined in the joint component to facilitate their instantiation. In other cases (especially as regards geometry), they have received a value when calling each component to avoid repeating too many variables. These data must

later be treated as variables when the partition is performed so that the correct value can be assigned in each case.

### 6.2. TIMING

The timing of the filtering cycle is achieved by a series of commands in the discrete part of the model.

This timing allows the state of the valves and the values of the parameters that indicate the change of stage to be set at any moment during the cycle (see table 2). Both phenomena are modelled with ZONE-type commands.

Valve status during the cycle						
Valve	Open valve stages					
	1	2	3	4	5	6
VA supply valve	x	x	x			
VD decompression valve				x	x	x
VE extraction valve				x	x	
VF filtering valve		x	x			
VN levelling valve					x	x
VRe recirculation valve				x	x	x
VRs recirculation valve	x					

Table 2. Valve status during the cycle

### 6.3 COMPONENT CODE

The final state of the 'filter' component is set out below:

```

-----
----- FILTRO -----
-----
COMPONENT Filtro_new_F IS_A cicloFiltro_new_F
  (SET_OF(Chemical) jug_mix, SET_OF(Chemical) ld_mix,
  BOOLEAN simple, INTEGER n_fallo1, INTEGER n_fallo2)

PORTS
  IN  analog_signal      s_in
  IN  jugo_depurado (Mix=jug_mix,
    simple=simple, ld=FALSE) jugo_in
  OUT jugo_depurado (Mix=jug_mix,
    simple=simple, ld=FALSE) jugo_out_ftr
  OUT jugo_depurado (Mix=jug_mix,
    simple=simple, ld=FALSE) jugo_out_rec
  OUT jugo_depurado (Mix=ld_mix,
    simple=simple, ld=TRUE) jugo_out_niv
  OUT jugo_depurado (Mix=ld_mix,
    simple=simple, ld=TRUE) lodo_out

DATA
  REAL t_espera = 0.    "tiempo de espera (s)"
  REAL t_ini          "tiempo comienzo ciclo"
  REAL ON = 1.        "parámetro: filtro funcionando"
  REAL Po = 1.5       "presión inicial en tela del filtro"
  REAL Pf = 1.1       "presión en la descompresión"
  REAL P0 = 1.
  REAL Wmax = 43.     "flujo máximo entra filtro (kg/s)"
  REAL f_esp = 50.    "factor corrección espesor torta filtración"

DECLS
  INTEGER ciclo        "nº de ciclos realizados"
  BOOLEAN New_ciclo    "parámetro comienzo de un ciclo"
  REAL Fin_ciclo       "parámetro indica final de ciclo"
  INTEGER paso         "pasos apertura/cierre válvulas"

```

```

REAL etapa_tela      "parámetro etapa tela del filtro"
REAL kk[3]           "parámetro de etapa en tela del filtro"
REAL descarga_cono1 "parámetro paso lodo cono1 a cono2"
REAL llenado_cono2  "parámetro llenado de cono 2"
REAL tf, aux        "tiempo transcurrido en cada ciclo"
REAL ff             "coeficiente eficacia filtrado (0-1)"
REAL ff2            "coeficiente simular colmatado (0-1)"
REAL t_p1,t_p2,t_p3,t_p4,t_p5,t_p6 "tiempo desde inicio ciclo"
REAL VA,VD,VE,VF,VN,VR_e,VR_s,VR "estado válvulas"

```

TOPOLOGY

```

depo_soplado (jugo_mix=jug_mix, simple=simple, ld=FALSE)
depo_soplado(Vmax=1.25, D=1.5, z_in1=0.)

```

```

tela_new_F(jug_mix=jug_mix, ld_mix=ld_mix, simple=simple)
tela(etapa=etapa_tela, Po=Po, Pf=Pf, k1=kk[1],
      k2=kk[2], k3=kk[3], Wmax=Wmax*ff2,
      f_esp=f_esp, ff=ff)

```

```

cono1_new_F(ld_mix=ld_mix, simple=simple)
cono1(descarga=descarga_cono1, vN_ON=VN)

```

```

cono2_new_F(ld_mix=ld_mix, simple=simple, C_dist=FALSE)
cono2(llenado=llenado_cono2,
      Fin_ciclo=Fin_ciclo)

```

```

valvula_filtro_new_F(jugo_mix=jug_mix, simple=simple,
ld=FALSE ,tipo=1)
valv_VRe(v_ON=VR_e,k=372.677)

```

```

valvula_filtro_new_F(jugo_mix=jug_mix, simple=simple,
ld=FALSE ,tipo=2)
valv_VRs(v_ON=VR_s,k=78.8266)

```

```

valvula_filtro_new_F(jugo_mix=ld_mix, simple=simple,
ld=TRUE, tipo=1)
valv_VE(v_ON=VE,k=996.)

```

```

valvula_filtro_new_F(jugo_mix=jug_mix, simple=simple,
ld=FALSE, tipo=2)
valv_VF(v_ON=VF,k=70.17)

```

```

CONNECT s_in      TO tela.s_in
CONNECT jugo_in   TO tela.jugo_in1
CONNECT tela.jugo_out2 TO valv_VRs.f_in
CONNECT valv_VRs.f_out TO depo_soplado.f_in1
CONNECT depo_soplado.f_out1 TO valv_VRe.f_in
CONNECT valv_VRe.f_out TO tela.jugo_in2
CONNECT tela.jugo_out1 TO valv_VF.f_in
CONNECT tela.lodo_out TO cono1.f_in
CONNECT cono1.f_out1 TO cono2.f_in
CONNECT cono2.f_out TO valv_VE.f_in
CONNECT valv_VE.f_out TO lodo_out
CONNECT depo_soplado.f_out1_rebose TO jugo_out_rec
CONNECT cono1.f_out2 TO jugo_out_niv
CONNECT valv_VF.f_out TO jugo_out_ftr

```

INIT

```

aux = 0.
tf = 0.
ciclo = 1
New_ciclo = TRUE   AFTER t_ini
descarga_cono1 = 0.
llenado_cono2 = 0.

```

DISCRETE

```

WHEN (ON==1 AND New_ciclo==TRUE) THEN
  New_ciclo=FALSE
  aux = TIME
  paso = 1
  paso = 2   AFTER t_p1
  paso = 3   AFTER t_p2
  paso = 4   AFTER t_p3
  paso = 5   AFTER t_p4
  paso = 6   AFTER t_p5
  paso = 7   AFTER t_p6

```

```

etapa_tela = 1.
etapa_tela = 2. AFTER t_p1
etapa_tela = 3. AFTER t_p3
descarga_cono1 = 1.   AFTER t_p6+2.
descarga_cono1 = 0.   AFTER t_p6+15.
llenado_cono2 = 1.    AFTER t_p6+3
llenado_cono2 = 0.    AFTER t_p6+4
ciclo = ciclo + 1     AFTER t_p6
Fin_ciclo = 1.        AFTER t_p6+20
Fin_ciclo = 0.        AFTER t_p6+21
New_ciclo = TRUE     AFTER t_p6 + t_espera

```

END WHEN

CONTINUOUS

```

---- tiempo desde inicio del ciclo hasta cada paso
t_p1 = t1_1
t_p2 = t_ciclo1 - t3_1 - t2_1
t_p3 = t_p2 + t2_1
t_p4 = t_p3 + t5_1
t_p5 = t_p3 + t4_1
t_p6 = t_ciclo1
---- estado de las válvulas
VA = ZONE(paso==1 OR paso==2 OR paso==3) 1.
OTHERS 0.
VD = ZONE(paso==4 OR paso==5 OR paso==6) 1.
OTHERS 0.
VE = ZONE(paso==4 OR paso==5) 1.
OTHERS 0.
VF = ZONE(paso==2 OR paso==3) 1.
OTHERS 0.
VN = ZONE(paso==5 OR paso==6) 1.
OTHERS 0.
VR_e = ZONE(paso==4 OR paso==5 OR paso==6) 1.
OTHERS 0
VR_s = ZONE(paso==1) 1.
OTHERS 0.
VR = ( VR_e + VR_s)*1.
---- etapas en la tela del filtro
kk[1] = ZONE(etapa_tela==1.) 1.
OTHERS 0.
kk[2] = ZONE(etapa_tela==2.) 1.
OTHERS 0.
kk[3] = ZONE(etapa_tela==3.) 1.
OTHERS 0.
---- tiempo transcurrido en cada ciclo
tf = ZONE(ON==1 AND TIME>=t_ini)
      (TIME - aux)
      OTHERS 0.
---- fallos
ff = ZONE (abs(fallo[n_fallo1])<0.000001) 1.
      OTHERS fallo[n_fallo1]/100.
ff2 = ZONE (abs(fallo[n_fallo2])<0.000001) 1.
      OTHERS fallo[n_fallo2]/100.
END COMPONENT

```

6.4. PARTITION AND EXPERIMENT

It is not possible to perform an experiment directly with the component alone because its parameters need to be determined. A second component where the filter call appears needs to be defined as well.

COMPONENT prueba\_filtro

```

TOPOLOGY
Filtro_new_F (tipo=1, jug_mix=jugoDep_simple ,
              ld_mix=lodo_simple, simple=TRUE,
              n_fallo1=429 , n_fallo2=430)
      filtro1(t_ini=0., ON=1., t_ciclo1=240., t1_1=30.,
              t2_1=5, t3_1=30., t4_1=6., t5_1=2.)
END COMPONENT

```

The variables that need to be set as boundary conditions for the performance of the experiment are those that define the inlet and outlet flows and correspond to the free ports. Furthermore, the data that need to be modified for the characterisation of the filter must be set as boundary conditions. Thus, the boundary conditions are:

- composition and temperature of the cloudy liquor flow entering the filter
- control signal from the level regulator of the cloudy liquor storage tank
- proportionality constants of the valves
- geometry of the elements that make up the filter
- pressures of the elements that make up the filter

## 7 RESULTS

The figures below represent the typical discontinuous behaviour of a filter, and are shown as an example of the results obtained by the simulation.

Figure 6 shows the sequencing of the valves that allows the timing of the filtering cycles.

Figure 7 shows the accumulation of mass in each of the elements that make up the filter: mesh, blowing tank, cone 1 and cone 2

Figure 8 shows the inlet and outlet flows from the filter: inlet of cloudy liquor from the blowing tank, filtered liquor that spills over from the blowing tank and levelling liquor (both are recirculated to the blowing tank), and finally filtered liquor and sludge flowing through to the following process stages.

Figure 9 shows the variation in the density of the sludge during its discharge, caused by the distribution of the sludge inside the cone.

Figure 10 shows the inlet and outlet flows to the blowing tank.

Filtering is a batch process, which complicates its simulation. The phenomena that occur are not linear, and the operating conditions vary considerably over time (filtering cycles). This means that the calculation model undergoes many changes throughout the simulation, causing a great numerical and mathematical complexity that grows exponentially whenever several filters are joined and the cleaning section is completed.

A filter model includes 463 equations and 463 variables (26 status variables). 28 variables need to be determined as boundary conditions to carry out an experiment.




Figure 6: Valve position (open: 1 ; closed: 0) during the filtering cycles.




Figure 7: Accumulated masses in each element of the filter (mesh: blue; blow\_tank: green; cone1: red; cone2: brown)




Figure 9: Detail of the density variation at the filter discharge (density: red; sludge discharge flow: blow)




Figure 8: Inlet and outlet flows from the filter during the cycle (cloudy liquor inlet: red; cloudy liquor outlet: brown; outlet liquor by levelling: green; liquor spills over from blowing tank: light blue; sludge outlet: dark blue)




Figure 10: Inlet and outlet flows from the blowing tank (inlet flow: red; spillover outlet flow: green; outlet flow for cake discharge: brown)

## 8 CONCLUSIONS

The model developed complies with all the requirements set out and correctly represents the real behaviour of the system.

The simulation tool used, Ecosimpro, has allowed the modelling of simpler elements which, once assembled, allow a more complex system to be simulated.

The main problem that arose during the simulation of the filters was to solve the mathematical equations that represent the physical model. These equations conditioned the choice of the theoretical model.

The final component obtained may be reused by simply changing the defining parameters (geometry and operating conditions).

These models will be of great use in the training of personnel because it is not necessary to know the code used to build them.

## Appendix

Components from other libraries used.

PORTS library	
<i>Component</i>	<i>Description</i>
analog_signal	Analogue signal port
jugo_depurado	Port for liquor or sludge in the cleaning section

FLOW ELEMENTS library	
<i>Component</i>	<i>Description</i>
valvula_filtro_new_F	Clean liquor valve

BASIC PROCESS ELEMENTS library	
<i>Component</i>	<i>Description</i>
deposito_jugoDepurado	Tank with an inlet and an outlet

## References

- [1] Z. Bubnk, P. Kadlec, D. Urban, M. Bruhns (1995) "Sugar Technologists Manual: chemical and physical data for sugar manufacturers and users" 8th edition. Bartens.
- [2] McGinnis (1971) "Beet-Sugar Tenhnology" 2d Edition. Beet sugar Development Foundation.
- [3] P.W. van der Poel, H. Schiweck, T. Schwart (1998) "Sugar Technology: Beet and Cane Sugar Manufacture" 8th edition. Bartens.